Journal of the Korea Society of Mathematical Education Series D:
Research in Mathematical Education
Vol. 10, No. 1, March 2006, 33–47

韓國數學教育學會誌 시리즈 D:
<數學教育研究>
제 10 권 제 1 호 2006 년 3 월, 33–47

Using Spreadsheets with Mathematically Gifted Students[1]

Arganbright, Deane
Visiting Professor, Department of Mathematics, KAIST, 373-1 Guseong-dong, Yuseong-gu, Daejeon 305-701, Korea; E-mail: arganbright@math.kaist.ac.kr

(Received March 14, 2006)

Finding good ways to support the further development of mathematically gifted students is a challenge for all mathematics educators. Simply moving able students on more rapidly to the next level of traditional mathematical instruction seems to be a limited approach, while providing supplementary enrichment material or specialized mathematical software requires us to ensure that doing so is truly worthwhile for the students. This paper presents an approach that the author has used with students of diverse capabilities in both technologically advanced and developing nations – investigating mathematical ideas using a spreadsheet.

Keywords: spreadsheets, graphics, animation, modeling, gifted education
ZDM Classification: U70
MSC2000 Classification: 97U70

INTRODUCTION

The use of spreadsheets by gifted students for the study and exploration of mathematics has strengths that argue well for it.

- A spreadsheet provides a non-traditional, yet natural and readily available, medium for developing mathematical insights.
- An increasing abundance of diverse, interesting, and valuable mathematical topics are available for spreadsheet study.
- Using a spreadsheet can make many mathematical ideas more interesting to students.
- A spreadsheet supplies diverse graphical capabilities for mathematics, giving students an opportunity to exhibit their creative visual skills.

[1] This paper will be presented at the Eleventh International Seminar of Mathematics Education on Creativity Development at the Chonam National University, Gwangju, Korea, April 7, 2006.

- The use of spreadsheet graphics, including animation, is both instructive and interesting in its own right.
- Students can discover that interesting mathematics appears in diverse areas of human experience – not only in science and engineering, but also in the arts, social sciences, cultural studies, and environmental issues.
- Teachers can incorporate spreadsheets into their current teaching, while encouraging more gifted students to discover and investigate new and challenging supplementary examples and concepts.

A spreadsheet approach has three more particularly compelling features.

- The actual process of creating mathematical spreadsheet models can itself be used to in develop mathematical concepts.
- Spreadsheet models make significant topics of a more advanced nature – many that in a traditional approach would be inaccessible – available to students in a mathematically sound manner
- The spreadsheet is the principal mathematical tool of the workplace – one that also is increasingly used by scientists and engineers. Its use enables students to acquire skills that are valued in the workplace and laboratory, and provides a good medium for a novel parent-student interface since they often use the same computer software

In this paper we provide four illustrative spreadsheet examples to indicate some of the range of topics and spreadsheet features that are available. We also include some graphic output from other spreadsheet models developed for other areas. Many additional examples will be found in the references.

We use the most popular spreadsheet program, *Microsoft Excel,* although others are applicable as well. Discussions of many more examples, together with a CD of *Excel* implementations and explanations of *Excel* operations, are available in (Neuwirth & Arganbright 2004). In this paper we briefly outline one graphic animation technique, the use of scroll bars. Others, such as the use of simple macros to create "movies" are discussed in (Arganbright 2005; Neuwirth & Arganbright 2004), with example *Excel* files that are available at each source. We provide a number of other printed and on-line references and sources of materials as well.

I. GEOMETRY

The field of geometry presents students with a broad range of mathematically interesting, challenging, and accessible topics to explore while encountering diverse new mathematical ideas and techniques. Here we present one example that also demonstrates a major spreadsheet animation technique.

There are two general approaches for using spreadsheets in mathematics. The first is to derive the mathematics in a standard way, and then to design and create an animated spreadsheet model that illustrates the development. The second is to use the spreadsheet creation process itself to develop the mathematics. In this example we employ the first approach to show how an ellipse is created if we tie a length of string to two pins and stretch it to find the set of all points for which the sum of the distances between it and the two fixed points is constant. Later we will show the output of a separate example that uses the second approach.

Suppose that we wish to create an ellipse whose foci are at the points $(-c,0)$ and $(c,0)$. These are the points at which we affix the string. In our example of Figure 1, we use $c=4$ together with a string that has length $2a=10$. We can use a standard derivation to show that the ellipse has a major semi-axis of length $a=5$, and a minor semi-axis of $b=\sqrt{a^2-c^2}=3$. Consequently, the curve can be generated from the parametric equations $x=a\cos t$, $y=b\sin t$, for $0\le t\le 2\pi$. Our model will generate the curve, incorporating a scroll bar to create an animated tracing of the ellipse. Our initial display is shown in Figure 1, with numerical output provided in Figure 2.



Figure 1

	A	B	C	D	E	F	G	H
1	a	5	b	3	c	4		
2	N	4	x	4.988	y	0.209		
3								
4	k	t	x	y	X	Y	xx	yy
5	0	0	5	0	5	0	4	0
6	1	0.017	4.999	0.052	4.999	0.052	4.988	0.209
7	2	0.035	4.997	0.105	4.997	0.105	-4	0
8	3	0.052	4.993	0.157	4.993	0.157		
9	4	0.07	4.988	0.209	4.988	0.209		
10	5	0.087	4.981	0.261	4.988	0.209		

Figure 2

In Figure 3 we begin by entering values for the parameters a and b in Cells B1 and D1, and then computing the value of c in Cell F1 as $\sqrt{a^2-b^2}$. Alternatively, we could enter the values of a and c, and then calculate b as $\sqrt{a^2-c^2}$. Next, down Column A we generate degrees, k, in steps of size 1, ranging from 0 to 360, and then use the *Excel* RADIANS function in Column B to convert to these to radians, t. In Columns C and D we compute $a\cos t$ and $b\sin t$. The references to the parameters a and b are absolute (shown by $), while the references for t are relative. At this stage we can use the mouse to select Columns C:D and use the chart wizard to create an xy-graph of the complete ellipse.

However, we also want to create an animation effect, so we use Cell B2 as a point counter, N, with N an integer that varies over the range $0 \le N \le 360$. Our display in Figure 2 shows the output for $N = 4$. In Columns E and F we use the built-in *Excel* IF function to provide the current value of (x, y) for those points $k \le N$, and otherwise to repeat the value of the cell above. The effect of this is to generate only those points in the range $0 \le k \le N$. We will use a scroll bar to repeatedly increase the value N in steps of size 1, generating additional points and thereby tracing out the curve.

	A	B	C	D	E	F	G	H
1	a	5	b	3	c	=SQRT(B1^2-D1^2)		
2	N	4	x	=VLOOKUP(B2,t,3)	y	=VLOOKUP(B2,t,4)		
3								
4	k	t	x	y	X	Y	xx	yy
5	0	=RADIANS(A5)	=$B$1*COS(B5)	=$D$1*SIN(B5)	=IF($A5<=$B$2,C5,E4)	=IF($A5<=$B$2,D5,F4)	=F1	0
6	=1+A5	=RADIANS(A6)	=$B$1*COS(B6)	=$D$1*SIN(B6)	=IF($A6<=$B$2,C6,E5)	=IF($A6<=$B$2,D6,F5)	=D2	=F2
7	=1+A6	=RADIANS(A7)	=$B$1*COS(B7)	=$D$1*SIN(B7)	=IF($A7<=$B$2,C7,E6)	=IF($A7<=$B$2,D7,F6)	=-G5	0

Figure 3

Here is what we do to produce the scroll bar. From the main command menu we select the options View, Toolbars, Control Toolbox. Then, from the resulting toolbar that is shown in Figure 4, we first click on the upper left button to enter the Design Mode, and then click on the scroll bar icon.



Figure 4

We next use the mouse to drag out a scroll bar and right click in it. From the menu that appears, set we designate Cell B2 as the linked cell for the scroll bar, and set 360 as its maximum value. We then close that dialog box and click to exit the Design Mode. Now our scroll bar is linked to Cell B2. As we move the slider to the right, the value of N increases to produce additional points in the graph, and students can see the curve as it is traced out.

To embellish our output further, we use the VLOOKUP function to generate the coordinates of the current point, N. Then, in Columns G:H we incorporate a new series into our graph to form the string by plotting the points $(-c,0)$, (x,y), and $(c,0)$, and showing both a line (string) and circle markers (pins, pencil). The result appears as our initial screen display of Figure 1. We enhance our visual output further by (x,y) coordinates to create the image of a pencil whose point is at the latest new (x,y) point. This is shown in Figure 5. As the slider in the scroll bar moves, so does the pencil.



Figure 5



Figure 6

The display in Figure 6 shows the output of another example that creates an ellipse, not from equations, but from the actual spreadsheet creation process. We leave the creation of this as an exercise. The mechanical device involved is called a *trammel*. A trammel is a solid rod with dowels at two points that travel in channels along the x- and y-axes. At one end of the rod is a pencil. Moving the pins in the channels causes the pencil to trace out a curve that we can verify is an ellipse. Students can find a great number of other designs of similar nature to explore, both in books and by surfing the Web.

We can use the same spreadsheet approach to produce animations of the graphs of a wealth of polar and parametric equations. In Figure 7 we show a fly tracing out the lissajous curve, $x = \cos 3t$, $y = \sin 5t$. Here we use linear algebra to rotate, scale, and translate the image of the fly so that it always moves tangent to the curve. Details of the construction are provided on-line in (Arganbright 2005). We can also graph polar equations by first computing $r = f(t)$, and then generating the x- and y-coordinates as $x = r\cos t$, $y = r\sin t$. In Figure 8 we create the graph of $r = \cos 2t$, tracing the development of the curve on top of the completed curve.



Figure 7



Figure 8



Figure 9

Another excellent project for gifted students is the creation of illustrations of classical geometric constructions, such as the pedal curve shown in Figure 9 (see Arganbright 1993 for further examples and details). We draw a tangent line from a point that moves along the blue deltoid, together with a line from a given fixed point that is perpendicular to the first line. The point of intersection of the two lines moves to generate a *pedal* curve. As we use a slider to move the point along the deltoid, we see the pedal being traced out.

II. EXPLORING FUNCTIONS

Spreadsheet graphics give students an unbounded capacity to create visual mathematical investigations of the behavior of functions. Moreover, it is possible to create functions and analyze them in ways that generally are not possible with graphing calculators. Again, students have the ability to animate their output in many creative ways. Here we examine just one of the myriad of concepts available, that of a horizontal translation of a function.

We begin by creating the graph of a function $y = f(x)$. In the layout in Figures 10-11 we use Column A to count points on the graph, with the counter, *n*, ranging from 0 to 200. We then enter an initial value for *x* in Cell B6, and generate further values of *x* down Column B in steps of size *dx*, where the step size is set in Cell B1. In this illustration we use increments of size $dx = 0.04$ and an initial value of $x_0 = -4$, so that the domain of our function becomes $-4 \leq x \leq 4$. We enter the formula for the function, which here is $f(x) = 0.5x^3 - x$, in Cell C6 and copy it down the column.

Next we set the size, *c*, of a translation in Cell E1, and compute values of $x + c$ down Column D. We then produce the values of $f(x+c)$ by copying the formula for the function from Cell C6 into Cell E6, and copy it down Column E. To get the graph of the function $g(x) = f(x+c)$ for the original domain, we drag Column E into the graph to produce the red curve in Figure 10. See the paper and related *Excel* files in (Arganbright 2005) for details of this process.

In our graph we add a dotted horizontal line to indicate how the graph of the original function is translated by the amount *c*. Since the optimal location for this line will vary with the function that we use, we enter a label number (here 153) into Cell B2, that causes the line to be drawn at the *y*-value of the 153rd point. The formulas in Columns F and G generate the endpoints of the dotted line.

What remains for us to do is to create a scroll bar and link it to a cell that varies the value of *c*. However, a scroll bar in *Excel* can only assume non-negative integer values. So we enter an integer in Cell F1, link the scroll bar to this cell, and let it vary from 0 to 500 in steps of size 1. We then compute the value of the horizontal translation size, *c*, in Cell E1 as $(250 - \text{F1})/100$. As a result, *c* varies from –2.5 to 2.5 in steps of size 0.01. As we move the slider of the scroll bar, we see the visual effect of the translation.

There are many possible aspects for students to investigate that we can motivate from this example, including the illustrations of a vertical translation and scaling multiples.



	A	B	C	D	E	F	G
1	dx	0 04		c	1 92	58	
2	label	153			c= 1.92		
3							
4							
5	n	x	f(x)	x+c	f(x+c)	xx	yy
6	0	-4.00	-28.00	-2.08	-2.42	2.12	2.64
7	1	-3.96	-27.09	-2.04	-2.20	0.20	2 64
8	2	-3 92	-26.20	-2.00	-2 00		
9	3	-3.88	-25.33	-1.96	-1.80		
10	4	-3.84	-24.47	-1.92	-1.62		
11	5	-3.80	-23.64	-1.88	-1.44		
12	6	-3.76	-22.82	-1.84	-1.27		
13	7	-3 72	-22.02	-1.80	-1 12		
14	8	-3.68	-21.24	-1.76	-0.97		
15	9	-3.64	-20.47	-1.72	-0.82		
16	10	-3.60	-19.73	-1.68	-0.69		
17	11	-3.56	-19.00	-1.64	-0.57		
18	12	-3 52	-18.29	-1 60	-0 45		
19	13	-3.48	-17.59	-1.56	-0.34		
20	14	-3.44	-16.91	-1.52	-0.24		
21	15	-3.40	-16.25	-1.48	-0.14		
22	16	-3.36	-15.61	-1.44	-0.05		
23	17	-3 32	-14.98	-1.40	0 03		

Figure 10

	A	B	C	D	E	F
1	dx	0.04		c	=(250-F1)/100	58
2	label	153			="c= "&TEXT(E1,"0.00")	
3						
4						
5	n	x	f(x)	x+c	f(x+c)	
6	0	-4	=0.5*B6^3-B6	=B6+$E$1	=0.5*D6^3-D6	
7	=1+A6	=B6+$B$1	=0.5*B7^3-B7	=B7+$E$1	=0.5*D7^3-D7	
8	=1+A7	=B7+$B$1	=0.5*B8^3-B8	=B8+$E$1	=0.5*D8^3-D8	

Figure 11

Students can also create drawings to represent 3-dimensional images and functions of two real variables by designing some perspective maps of ¡ ^{3} in ¡ ^{2}, following schemes used in standard text books. One example is shown in Figure 12. In addition, they can use two graphs of ¡ ^{2} to create represent the domain and image for complex functions! Figure 13 provides an illustration of the output of the complex reciprocal function $f(z)=1/z$ on a grid of horizontal and vertical lines (see Arganbright 1993). This topic provides an excellent example of an area in which gifted students can investigate historically important concepts that is often not contained in today's curriculum.



Figure 12



Figure 13

III. MATHEMATICAL MODELING

Perhaps the area of spreadsheet mathematics that provides the greatest potential for gifted students lies in mathematical modeling. Not only is this an important topic, but provides teachers with a compelling way to introduce a wide range of ideas that demonstrate why mathematics is both useful and interesting. Moreover, in doing mathematical modeling, the actual process of creating a spreadsheet model itself can lead students to learning and developing new mathematics. This can be seen in the many examples found in (Neuwirth & Arganbright 2004). Topics from such diverse areas as genetics, planetary motion, heat flow, population growth, business inventories, medicine dosages, and legislative apportionment provide an abundance of topics to pursue. Spreadsheets often permit students to pursue these topics effectively even without needing any background from the traditional prerequisite fields of calculus and differential equations.

Here we consider the basic SIR model for the spread of an epidemic. In this model, we examine an isolated community of a fixed size. Residents are classified into three categories. Susceptible (S), who are free of the disease but contract it if they come into contact an Infected (I), who currently has the disease. We assume that the disease is not fatal, and that after a period, a certain percentage of the Infected are cured from the disease, and become Removals (R) who no longer can get the disease.

In our implementation in Figures 14-15 the population of the community is 2000, and the initial number of those infected is 3. During a certain time period, say a month, the people in the community interact with others at random, and the probability, p, that they contact any one other person in that time is assumed to be known. Here we use

$p = 0.001$. If a Susceptible comes into contact with an Infected, then he becomes infected. Thus, if in a given period there are k Infected, then to stay free of becoming infected, a Susceptible must avoid all k, with probability $(1-p)^k$, so that probability of becoming infected is $1-(1-p)^k$.

To create our model, we use the first column to count periods, enter the initial number of Infected (3) in Cell C4 and Removals (0) in Cell D4, and compute the number susceptible in Cell B4. The number of newly infected in the first period is found in Cell E4 as the product of the probability of meeting an infected with the number who are susceptible. The number of new removals is the product of the healing rate and the current number infected. We use a healing rate of 1, but treating it as a parameter, we can use any probability that is desired. In the next row, we compute the number susceptible in next period by subtracting the number of newly infected from the previous susceptible number. The total infected is found by adding the number newly infected to the previously infected and subtracting the newly cured. The total number of removals then is the sum of the old number and the new ones. After completing each initial computation, all that we need to do is to copy the formulas down their respective columns. In the formulas of Figure 14, only the cure and contact rates are absolute references.

	A	B	C	D	E	F
1	popn	2000	cnt rate	0.001	cure rate	1
2						
3	period	susceptible	infected	removals	new infect	new remove
4	0	=B1-C4	3	0	=B4*(1-(1-$D$1)^C4)	=C4*$F$1
5	=1+A4	=B4-E4	=C4+E4-F4	=D4+F4	=B5*(1-(1-$D$1)^C5)	=C5*$F$1
6	=1+A5	=B5-E5	=C5+E5-F5	=D5+F5	=B6*(1-(1-$D$1)^C6)	=C6*$F$1

Figure 14

In Figure 15 we see the first few rows of output.

	A	B	C	D	E	F
1	popn	2000	cnt rate	0.001	cure rate	1
2						
3	period	susceptible	infected	removals	new infect	new remove
4	0	1997.0	3.0	0.0	6.0	3.0
5	1	1991.0	6.0	3.0	11.9	6.0
6	2	1979.1	11.9	9.0	23.4	11.9
7	3	1955.7	23.4	20.9	45.3	23.4

Figure 15

The use of graphs greatly enhances this model. The graph of Figure 16 allows us to see what occurs. In our model the epidemic eventually dies out before everyone becomes infected. We can carry out a wide variety of investigations by including sliders for such aspects of the model the population size, the cure rate, the initial number of those who are infected, and the contact probability.

One reason that creating such a model is an excellent task for gifted students is that it allows them to consider why our model does not generate the epidemic "waves" that real diseases tend to produce. In considering this question, they may come up with several additional features that we should include in a model: new people move into or are born into the community, while others die, or perhaps the immunity to the disease only lasts for a limited period. Each of these considerations provides an excellent aspect to model, and including them in the models can be challenging. In Figure 17 we see a graph that is produced if immunity lasts for 12 months and the monthly cure rate is 85%. Note that now a new epidemic wave comes along about every 2.5 years.



Figure 16



Figure 17

IV. MATHEMATICAL ALGORITHMS

The spreadsheet is an excellent tool for implementing many mathematical algorithms that are iterative, recursive, or fit into a table format (see Arganbright 1985; McLaren 2004; Neuwirth & Arganbright 2004). A spreadsheet also allows students to create superior graphical representations that allow them to gain additional insights from the visualization. Among the advantages of using spreadsheets rather than other software, is that frequently a spreadsheet implementation can be created in exactly the same way that we would do by hand, so that the creation of a spreadsheet model itself reinforces the mathematics underlying the algorithm. Moreover, once the spreadsheet model is created, students can use it to interrogate the model by investigating the effects of changes in parameters, initial conditions, and functions. Fields such as algebra, calculus, numerical analysis, and statistics abound with stimulating examples. To find these examples, students can search books or surf the Web productively to find important and interesting topics and then design their own implementations. In most instances, there will be many different ways to proceed, providing students with boundless opportunities to develop their creativity skills.

Here we examine one such algorithm, the *secant method*, for locating the zeroes of a continuous real-valued function. This algorithm is similar to Newton's method (which itself provides another excellent subject to study) from calculus, but it does not require the knowledge of calculus or the use of derivatives.

The idea of the secant method can be seen using the graph shown in Figure 18. We take two estimates of a zero, x_0 and x_1, and find the line that passes through the points $(x_0, f(x_0))$ and $(x_1, f(x_1))$. The value, x_2, where the line crosses the x-axis is generally (but not always) a better approximation for the zero. We then repeat the same process using x_1 and x_2, and continue until we locate a zero to the degree of accuracy that we desire.



Figure 18



Figure 19

Before implementing the algorithm, we first note that the slope, m, of the line determined by the two points is given by $m = (y_1 - y_0)/(x_1 - x_0)$. The line then satisfies the equation $y - y_1 = m(x - x_1)$, so that we obtain the point x_2 by setting the value of y to 0 and x to x_2 to obtain $-y_1 = m(x_2 - x_1)$. We next solve for x_2 to get $x_2 = x_1 - y_1 / m$.

The spreadsheet layout for this algorithm is provided in Figures 20-21 using the function $f(x) = 0.4x^3 - 5x + 2$. We employ the first column to count iterations, n. We enter 0 and 1 in the first two cells in Column B as our initial estimates, compute $y = f(x)$ in next the column, and the slope, m, in the final column, starting from the second row and using the two most current points. In Cell B4 we enter the formula for x_2 that we developed above. We then copy each of these formulas down its column. All of the cell references are relative. We notice that in this case the algorithm converges quickly to the closest zero. This will not necessarily be the case, as is shown in Figure 22 using Newton's method (Neuwirth & Arganbright 2004). In designing an interactive model, it helps us if we plot the graph of the function (Figure 19) using techniques above, so that we know where to look for the zeroes.

	A	B	C	D
1	n	x	y	slope
2	0	0	2	
3	1	1	-2.6	-4.6
4	2	0.434783	-0.14104	-4.35047
5	3	0.402364	0.014238	-4.78965
6	4	0.405336	-4.4E-05	-4.80429
7	5	0.405327	-1.3E-08	-4.80285
8	6	0.405327	1.18E-14	-4.80285

Figure 20

	A	B	C	D
1	n	x	y	slope
2	0	0	=0.4*B2^3-5*B2+2	
3	=1+A2	1	=0.4*B3^3-5*B3+2	=(C3-C2)/(B3-B2)
4	=1+A3	=B3-C3/D3	=0.4*B4^3-5*B4+2	=(C4-C3)/(B4-B3)
5	=1+A4	=B4-C4/D4	=0.4*B5^3-5*B5+2	=(C5-C4)/(B5-B4)
6	=1+A5	=B5-C5/D5	=0.4*B6^3-5*B6+2	=(C6-C5)/(B6-B5)
7	=1+A6	=B6-C6/D6	=0.4*B7^3-5*B7+2	=(C7-C6)/(B7-B6)
8	=1+A7	=B7-C7/D7	=0.4*B8^3-5*B8+2	=(C8-C7)/(B8-B7)

Figure 21

From such a model students can experiment with their initial choices for x_0 and x_1 and discover patterns, and perhaps encounter some surprising results, because the algorithm does not always converge and may even oscillate. Sometimes it converges to an unexpected zero away from the initial estimate, as in the diagram for Newton's method in Figure 22. The reasons for these behaviors sometimes can be deduced from the investigations. In addition, students can pursue such other zero-finding methods as Newton's, false position, bisection, and fixed-point.



Figure 22

Most of the algorithms from numerical analysis (Arganbright 1985; McLaren 1997) also make excellent spreadsheet topics, and they frequently can be implemented with only a modest knowledge of calculus. Such topics include numerical integration, numerical differentiation, power series, numerical solutions of differential equations, iterative solutions of systems of linear equations, eigenvalues, and many more areas of computational mathematics.

V. ADDITIONAL TOPICS

In addition to the examples discussed above, here we provide a few graphs illustrating other topics which support productive topics for gifted students to pursue. For example, one can design interactive, animated models to illustrate the concepts of the δ-ε definition of limits (Figure 23) and the approximation of the area under a curve by rectangles (Figure 24).



Figure 23



Figure 24

In addition, artistically-inclined students can design a surprising range of traditional and cultural graphic images, as well as designing those of their own, as seen in Figures 25-27 (Arganbright 1993).



Figure 25



Figure 26



Figure 27

Additional spreadsheet applications abound in such disciplines as statistics, art, calculus, differential equations, design, numerical analysis, three-dimensional graphics, business, economics, science, ecology, environmental studies, graphic design, social studies, recreation, and fractals. We provide sources of some of these in the references.

REFERENCES

Arganbright, D. (2005): Enhancing Mathematical Graphical Displays in Excel through Animation," *Spreadsheets in Education*, **2(1)**, 125-147.
Available from http://www.sie.bond.edu.au/classresource.asp?id=10

Arganbright, D. (2005): "Developing Mathematics Creativity with Spreadsheets," *Journal of the Korea Society of Mathematical Education, Series D: Research in Mathematical Education*, **9(3)**, 187-201.

Arganbright, D. (2002): "Creative Spreadsheet Graphics in Mathematics Teaching and Modeling", *Technology in Mathematics Teaching*, Schriftenreihe Didaktik der Mathematik, **26**, Oesterreichischer Bundesverlag, Vienna, 153-156.

Arganbright, D. (1993): *Practical Handbook of Spreadsheet Curves and Geometric Constructions*, CRC Press, Boca Raton, Florida.

Arganbright, D. (1985): *Mathematical Applications of Electronic Spreadsheets*, McGraw-Hill, New York.

Hill, David & Lila Roberts. *Demos with Positive Impact.*
Available from http://mathdemos.gcsu.edu/morelinks2.html

McLaren, David (2004): *Spreadsheets and Numerical Analysis*. Chartwell-Bratt, 1997.

Neuwirth, Erich & Arganbright, D.: *The Active Modeler: Mathematical Modeling with Microsoft Excel*, Brooks-Cole.

Neuwirth, Erich: *Spreadsheets, Mathematics, Science, and Statistics Education.*
Available from Internet: http://sunsite.univie.ac.at/Spreadsite/

Sugden, Steve (ed.): *Spreadsheets in Education.*
Available from Internet:http://www.sie.bond.edu.au/